TB 13.8 Rotating Star 01-20-18

N. T. Gladd

Initialization: Be sure the file *NTGUtilityFunctions.m* is in the same directory as that from which this notebook was loaded. Then execute the cell immediately below by mousing left on the cell bar to the right of that cell and then typing “shift” + “enter”. Respond “Yes” in response to the query to evaluate initialization cells.

```
SetDirectory[NotebookDirectory[]];
(* set directory where source files are located *)
Get["NTGUtilityFunctions.m"]; (* Load utilities package *)
```

Purpose

This is the 3rd in a series of notebooks in which I work through material and exercises in the magisterial new book *Modern Classical Physics* by Kip S. Thorne and Roger D. Blandford. If you are a physicist of any ilk, BUY THIS BOOK. You will learn from a close reading and from solving the exercises. See TB 13.4 Polytropes 01-01-18 for more discussion of this project.

Exercise 13.8 *Problem: Rotating Planets, Stars and Disks*
Consider a stationary, axisymmetric planet star or disk differentially rotating under the action of a gravitational field. In other words, the motion is purely in the azimuthal direction.

(a) Suppose that the fluid has a *barotropic* equation of state $P = P(\rho)$. Write down the equations of hydrostatic equilibrium including the centrifugal force in cylindrical polar coordinates. Hence show that the angular velocity must be constant on surfaces of constant cylindrical radius. This is called von Zeipel's theorem. (As an application, Jupiter is differentially rotating and therefore might be expected to have similar rotation periods at the same latitude in the north and the south. This is only roughly true, suggesting that the equation of state is not completely barotropic.)

(b) Now suppose that the structure is such that the surfaces of constant entropy per unit mass and angular momentum per unit mass coincide.(This state of affairs can arise if slow convection is present.) Show that the Bernoulli function [Eq. (13.50)] is also constant on these surfaces. (Hint: Evaluate ∇B.)

Analysis and solution

Part (a)
Consider a rotating axi-symmetric star



and describe it’s dynamics using a cylindrical coordinate system.



Force balance on a fluid element at P involves the gravitational force and centrifugal force forces on fluid element



$$\frac{\nabla P(R,\ z)}{\rho(R,\ z)} = \boldsymbol{F_G}(R,\ z) + \boldsymbol{F_C}(R,\ z) = -\frac{G\,M}{R^2 + z^2}\,\mathbb{1}_r + \Omega(R,\ z)^2\,R\,\mathbb{1}_R \equiv \boldsymbol{G}_{\text{eff}}(R,\ z) \qquad (1)$$

where it has been assumed that the stellar mass is concentrated at the origin. Define the right hand side to be an "effective" gravitational force. Can the centrifugal force be represented as a potential? That requires that it be a conservative force, one for which the curl vanishes.

$$\nabla \times \boldsymbol{F_C}(R,\ z) = \nabla \times \Omega(R,\ z)^2\,R\,\mathbb{1}_R = 0 \qquad (2)$$

In detail

```
Curl[{Ω[R, z]^2 R, 0, 0}, {R, ϕ, z}, "Cylindrical"]
```

```
{0, 2 R Ω[R, z] Ω^(0,1)[R, z], 0}
```

If the centrifugal forces is conservative, Ω cannot depend on z.

$$\Omega = \Omega(R) \qquad (3)$$

which corresponds to rigid rotation with constant angular momentum on cylindrical surfaces. A representative surface with streamlines is shown



In this case

$$\boldsymbol{G}_{\text{eff}}(R,\ z) = -\nabla\ \Phi_{\text{eff}}(R,\ z) = \frac{\partial \Phi_{\text{eff}}(R,\ z)}{\partial R}\ \mathbb{1}_R + \frac{\partial \Phi_{\text{eff}}(R,\ z)}{\partial z}\ \mathbb{1}_z \tag{4}$$

where, in particular,

$$\Phi_{\text{eff}}(R,\ z) = -\frac{G\,M}{\sqrt{R^2 + z^2}} + \frac{\Omega(R)^2\,R}{2} \tag{5}$$

Then

$$\frac{\nabla P(R,\ z)}{\rho(R,\ z)} = \boldsymbol{G}_{\text{eff}}(R,\ z) = -\nabla\ \Phi_{\text{eff}}(R,\ z) \tag{6}$$

implies that the effective gravitational field is everywhere orthogonal to the surfaces of constant pressure (isobars). For some nominal parameters the isobars and field lines are shown (details of plot generation below). Note that this result is independent of choice of equation of state.



Figure 1) Nominal parameters are M = 1, G = 1, Ω = 2

Next, note that

$$\frac{\nabla P(R,\ z)}{\rho(R,\ z)} = \frac{1}{\rho(R,\ z)}\frac{\partial P(R,\ z)}{\partial R}\mathbb{1}_R + \frac{1}{\rho(R,\ z)}\frac{\partial P(R,\ z)}{\partial z}\mathbb{1}_z = -\frac{\partial \Phi_{eff}(R,\ z)}{\partial R}\mathbb{1}_R - \frac{\partial \Phi_{eff}(R,\ z)}{\partial z}\mathbb{1}_z \tag{7}$$

and further that

$$dP(R,\ z) = \frac{\partial P(R,\ z)}{\partial R}dR + \frac{\partial P(R,\ z)}{\partial z}dz \tag{8}$$

So

$$\frac{dP(R,\ z)}{\rho(R,\ z)} = \frac{1}{\rho(R,\ z)}\frac{\partial P(R,\ z)}{\partial R}dR + \frac{1}{\rho(R,\ z)}\frac{\partial P(R,\ z)}{\partial z}dz = -\frac{\partial \Phi_{eff}(R,\ z)}{\partial R}dR - \frac{\partial \Phi_{eff}(R,\ z)}{\partial z}dz = -d\Phi_{eff}(R,\ z) \tag{9}$$

or

$$\frac{dP(R,\ z)}{\rho(R,\ z)} = -d\Phi_{eff}(R,\ z) \tag{10}$$

This expression indicates that the surfaces for which $\Phi_{eff}(R, z)$ = constant (equipotentials) are also surfaces for which P(R, z) = constant (isobars).

As to von Zeipel's theorem. If a barotropic equation of state is assumed

$$P(R,\ z) = P(\rho(R,\ z)) \quad \Leftrightarrow \quad \rho(R,\ z) = \rho(P(R,\ z)) \tag{11}$$

Then

$$\frac{\mathrm{dP}(R,\ z)}{\rho(R,z)} = \frac{\partial P(\rho)}{\partial \rho}\,\frac{\mathrm{d}\rho(R,\ z)}{\rho(R,\ z)} = -\,\mathrm{d}\Phi_{\mathrm{eff}}(R,\ z) \tag{12}$$

Thus, surfaces for which ρ(R, z) = constant (isopycnics) also correspond to equipotentials.

or

$$\frac{1}{\rho(R,z)} = -\frac{\mathrm{d}\Phi_{\mathrm{eff}}}{\mathrm{dP}(\Phi_{\mathrm{eff}})} \quad \Leftrightarrow \quad \rho(R,\ z) = \rho(\Phi_{\mathrm{eff}}) \tag{13}$$

For a barotropic equation of state, the constant density surfaces (isopycnics) correspond to the constant pressure surfaces (isobars) and the constant potential surfaces (equipotentials).

I feel the BT's statement of part (a) of the problem is somewhat misleading by implying that the barotropic equation of state is the key assumption.

(a) Suppose that the fluid has a *barotropic* equation of state $P = P(\rho)$. Write down the equations of hydrostatic equilibrium including the centrifugal force in cylindrical polar coordinates. Hence show that the angular velocity must be constant on surfaces of constant cylindrical radius. This is called von Zeipel's theorem. (As an application,

To me, the logic is that the assumption of a conservative form for the centrifugal force by itself leads to both constant angular moment on cylindrical surfaces and that isobars coincide with equipotentials. Assuming a barotropic equation of state just leads to the additional result that isopycnics also coincide with the equipotentials.

Confused about this, I did some web research and found the following.

In the paper Effects of Stellar Rotation on p-mode frequencies, M. J. Goupil, in the book The Rotation of the Sun and Stars, edited by J-P Rozelot, I found the succinct statement

The star keeps it spherical symmetry. This happens for a uniform rotation or for a cylindrical rotation $d\Omega/dz = 0$ in a cylindrical coordinate system [141]. The von Zeipel [145, 146] law states that for a barotrope and a conservative rotation law, isobars and isopycnics coincide with the level surfaces (constant potential):

Part (b)

The Bernoulli function is

$$B = \frac{v^2}{2} + h + \Phi_G \tag{14}$$

For the case of rigidly rotating cylinders v = Ω R, and the first and third terms can be combined into the effective potential defined in Part(a).

$$B = \frac{(\Omega R)^2}{2} + h + \Phi_G = h + \Phi_{\text{eff}} \tag{15}$$

Along a streamline

$$\text{dB} = \boldsymbol{v} \cdot \nabla B$$

B will be a constant of the surfaces of constant angular momentum if dB = 0, and that requires that ∇B = 0

$$\nabla B = \nabla h + \nabla \Phi_{\text{eff}} = \nabla h \tag{16}$$

since $\nabla \Phi_{\text{eff}} = 0$ is known from part (a). So — is ∇h = 0?

Consider the gradient of h

$$\nabla h = \frac{\partial h}{\partial x} \mathbb{1}_x + \frac{\partial h}{\partial y} \mathbb{1}_y + \frac{\partial h}{\partial z} \mathbb{1}_z \tag{17}$$

From thermodynamics, h = h(P, s). Further, for a barotropic equation of state P = P(ρ) or ρ = ρ(P). Thus h = h(ρ, s).

$$\text{dh} = T\,\text{ds} + \frac{\text{dP}}{\rho} \tag{18}$$

If it is given that entropy is constant on a surface of constant angular momentum, then ds = 0 and

$$h = h(\rho) \tag{19}$$

For the barotropic equation of state, h(ρ) is defined as

$$h(\rho) \equiv \int \frac{\text{dP}(\tilde{\rho})}{\tilde{\rho}} = \int \frac{\partial P(\tilde{\rho})}{\partial \tilde{\rho}} \frac{d\tilde{\rho}}{\tilde{\rho}} \tag{20}$$

which is sometimes called the “pressure potential.” Then

$$\frac{\partial h}{\partial \rho} = \frac{1}{\rho} \frac{\partial P}{\partial \rho} \tag{21}$$

With this expression

$$\nabla h = \frac{\partial h}{\partial \rho}\frac{\partial \rho}{\partial x}\mathbb{1}_x + \frac{\partial h}{\partial \rho}\frac{\partial \rho}{\partial y}\mathbb{1}_y + \frac{\partial h}{\partial \rho}\frac{\partial \rho}{\partial z}\mathbb{1}_z$$
$$= \frac{1}{\rho}\frac{\partial P}{\partial \rho}\frac{\partial \rho}{\partial x}\mathbb{1}_x + \frac{1}{\rho}\frac{\partial P}{\partial \rho}\frac{\partial \rho}{\partial y}\mathbb{1}_y + \frac{1}{\rho}\frac{\partial P}{\partial \rho}\frac{\partial \rho}{\partial z}\mathbb{1}_z$$
$$= \frac{1}{\rho}\frac{\partial P}{\partial x}\mathbb{1}_x + \frac{1}{\rho}\frac{\partial P}{\partial y}\mathbb{1}_y + \frac{1}{\rho}\frac{\partial P}{\partial z}\mathbb{1}_z$$
$$= \frac{1}{\rho}\nabla P$$

In Part(a) it was shown that isobars and surfaces of constant angular momentum coincide. Thus

$$\nabla h = \frac{1}{\rho}\nabla P = 0 \qquad (23)$$

and B is shown to be a constant on surfaces of constant angular momentum.

Appendix A - Construction of visualizations

The rotating star

```
Module[{O = {0, 0, 0}, 𝟙x = {1, 0, 0}, 𝟙y = {0, 1, 0},
  𝟙z = {0, 0, 1}, range, axes, rotationArrow, CyltoCar, Axis, G},
 CyltoCar[R_, θ_, z_] := CoordinateTransform[
   "Cylindrical" → "Cartesian", { R, θ, z}];
 Axis[OO_, P_, lab_, mult_, color_] :=
  {Directive[color, Thick], Line[{OO, P}], Text[Stl[lab], mult P]};
 axes = With[{mult = 1.2, color = White},
   Axis[O, #[[1]], #[[2]], mult, color] & /@ {{𝟙x, "x"}, {𝟙y, "y"}, {𝟙z, "z"}}];
 range = 1.2 {{-1, 1}, {-1, 1}, {-1, 1}};
 rotationArrow = With[{R = 0.2, θ = π / 4, z = 1.1, δθ = π / 4},
   {White, Arrowheads[0.025], Arrow[{CyltoCar[R, θ, 1 z], CyltoCar[R, θ + δθ, z]}],
    Stl@Text["Ω", {R + 0.1, 0, z - 0.1}]}];
 G[1] = Graphics3D[{axes, rotationArrow}, BoxRatios → Automatic, Boxed → False,
   Background → Black, Axes → None, PlotRange → range, ViewPoint → {2, 2, 1} ];
 G[2] = With[{α = 0.5, R = 0.6}, ContourPlot3D[{α (x^2 + y^2) + z^2 == R^2}, {x, -1, 1},
    {y, -1, 1}, {z, -1, 1}, Mesh → False, PlotLabel → Stl["ρbulge (R, z)"],
    ContourStyle → Directive[Yellow, Opacity[0.75], Specularity[White, 30]],
    BoxRatios → Automatic, Boxed → False, Background → Black,
    Axes → None, PlotRange → range, ViewPoint → {2, 2, 1}]];
 (* This is a way of placing the circle indicating rotation *)
 G[3] =
  ParametricPlot3D[{0.2 Cos[ϕ], 0.2 Sin[ϕ], 1.1 }, {ϕ, 0, 2 π}, PlotStyle → White,
   Axes → None, PlotRange → range, ViewPoint → {2, 2, 1}, Background → Black];
 Show[G[1], G[2], G[3]]]
```



Isobars and streamlines representing gravitational field

$$g_{\text{eff}} = \left(-\frac{\partial \Phi_G}{\partial R} + \Omega^2 R\right)\hat{\mathbf{1}}_R - \frac{\partial \Phi_G}{\partial z}\hat{\mathbf{1}}_z$$

```
wA[1] = { -D[ΦG[R, z], R] + Ω^2 R, - D[ΦG[R, z], z]}
```

$$\left\{R\,\Omega^2 - \Phi G^{(1,0)}[R, z],\ -\Phi G^{(0,1)}[R, z]\right\}$$

```
wA[2] = wA[1] /. ΦG → Function[{R, z}, - G M / Sqrt[R^2 + z^2]]
```

$$\left\{-\frac{G\,M\,R}{\left(R^2+z^2\right)^{3/2}} + R\,\Omega^2,\ -\frac{G\,M\,z}{\left(R^2+z^2\right)^{3/2}}\right\}$$

```
Clear[gEff, ΦEff];
gEff[R_, z_, Ω_, G_, M_] := {- G M R / (R^2 + z^2)^(3/2) + R Ω^2, - G M z / (R^2 + z^2)^(3/2)};

ΦEff[R_, z_, Ω_, G_, M_] := - G M / Sqrt[R^2 + z^2] - R^2 Ω^2 / 2;
```

```
-Grad[ΦEff[R, z, Ω, G, M], {R, z}]
```

$$\left\{-\frac{G\,M\,R}{\left(R^2+z^2\right)^{3/2}} + R\,\Omega^2,\ -\frac{G\,M\,z}{\left(R^2+z^2\right)^{3/2}}\right\}$$

effective gravity is everywhere orthogonal to isobars

```
Module[{M = 1, G = 1, RMax = 0.5, zMax = 0.5, image = 350,
  range = {{0, 0.5}, {0, 0.5}}, left, right, bottom, top, GG},
 {{left, right}, {bottom, top}} = {{Rotate[Stl["z"], -π/2], ""}, {Stl["R"],
    Stl["Isobars and stream lines indicating direction of effective gravity"]}};
 GG[1] = With[{Ω = 2}, StreamPlot[gEff[R, z, Ω, G, M], {R, 0, RMax},
    {z, 0, zMax}, StreamStyle → LightGray, PlotRange → range,
    ImageSize → image, FrameLabel → {{left, right}, {bottom, top}}]];
 GG[2] = With[{Ω = 2}, ContourPlot[ΦEff[R, z, Ω, G, M], {R, 0, RMax},
    {z, 0, zMax}, PlotRange → range, ImageSize → image,
    ContourShading → None, FrameLabel → {{left, right}, {bottom, top}}]];
 Overlay[{GG[2], GG[1]}]]
```



Illustration of representative surface of constant angular momentum

```
Module[{O = {0, 0, 0}, 𝟙x = {1, 0, 0}, 𝟙y = {0, 1, 0}, 𝟙z = {0, 0, 1},
  range, axes, rotationArrow, streamLineArrows, CyltoCar, Axis, G},
 CyltoCar[R_, θ_, z_] := CoordinateTransform[
   "Cylindrical" → "Cartesian", { R, θ, z}];
 Axis[OO_, P_, lab_, mult_, color_] :=
  {Directive[color, Thick], Line[{OO, P}], Text[Stl[lab], mult P]};
 axes = With[{mult = 1.2, color = White},
   Axis[O, #[[1]], #[[2]], mult, color] & /@ {{𝟙x, "x"}, {𝟙y, "y"}, {𝟙z, "z"}}];
 range =  1.2 {{-1, 1}, {-1, 1}, {-1, 1}};
 rotationArrow = With[{R = 0.2, θ = π / 4, z = 1.1, δθ = π / 4},
   {White, Arrowheads[0.025], Arrow[{CyltoCar[R, θ, 1 z], CyltoCar[R, θ + δθ, z]}],
    Stl@Text["Ω", {R + 0.1, 0,  z - 0.1}]}];
 streamLineArrows = With[{R = 0.5, θ = 3 π/8, z = 0.25, δθ = π/8},
   {White, Arrowheads[0.025], Arrow[{CyltoCar[R, θ, z], CyltoCar[R, θ + δθ, z]}],
    Arrow[{CyltoCar[R, θ, 0], CyltoCar[R, θ + δθ, 0]}],
    Arrow[{CyltoCar[R, θ, -z], CyltoCar[R, θ + δθ, -z]}]}];
 G[1] = Graphics3D[{axes, rotationArrow, streamLineArrows},
   BoxRatios → Automatic, Boxed → False, Background → Black,
   Axes → None, PlotRange → range, ViewPoint → {2, 2, 1} ];
 G[2] = With[{α = 0.5, R = 0.6}, ContourPlot3D[{α (x^2 + y^2) + z^2 == R^2}, {x, -1, 1},
    {y, -1, 1}, {z, -1, 1}, Mesh → False, PlotLabel → Stl["ρbulge(R, z)"],
    ContourStyle → Directive[Yellow, Opacity[0.25], Specularity[White, 30]],
    BoxRatios → Automatic, Boxed → False, Background → Black,
    Axes → None, PlotRange → range, ViewPoint → {2, 2, 1}]];
 G[3] = With[{R = 0.5}, ContourPlot3D[{ (x^2 + y^2) == R^2}, {x, -1, 1}, {y, -1, 1},
    {z, -1, 1}, Mesh → False, PlotLabel → Stl["ρbulge(R, z)"], ContourStyle →
     Directive[Lighter[Red, 0.5], Opacity[0.75], Specularity[White, 30]],
    BoxRatios → Automatic, Boxed → False, Background → Black,
    Axes → None, PlotRange → range, ViewPoint → {2, 2, 1}]];

 (*streamLines = {Red, {0.2 Cos[ϕ],0.2 Sin[ϕ],1.1 }};*)
 (* This is a way of placing the circle indicating rotation *)
 G[4] =
  ParametricPlot3D[{{0.2 Cos[ϕ], 0.2 Sin[ϕ], 1.1 }, {0.5 Cos[ϕ], 0.5 Sin[ϕ], -0.25 },
    {0.5 Cos[ϕ], 0.5 Sin[ϕ], 0}, {0.5 Cos[ϕ], 0.5 Sin[ϕ], 0.25 }},
   {ϕ, 0, 2 π}, PlotStyle → White, Axes → None, PlotRange → range,
   ViewPoint → {2, 2, 1}, Background → Black];
 Show[G[1], G[2], G[3], G[4]]]
```



Cylindrical geometry

```
Module[{O = {0, 0, 0}, 𝟙x = {1, 0, 0}, 𝟙y = {0, 1, 0}, 𝟙z = {0, 0, 1},
  axes, range, P, Q, ϕArc, refLines, CyltoCar, Axis, ArcArrow3D, G},
 CyltoCar[R_, θ_, z_] := CoordinateTransform[
   "Cylindrical" → "Cartesian", { R, θ, z}];
 Axis[OO_, P_, lab_, mult_, color_] :=
  {Directive[color, Thick], Line[{OO, P}], Text[Stl[lab], mult P]};
 ArcArrow3D[R_, θS_, θF_, z_] :=
  {Arrow@Table[CyltoCar[R, θ, z], {θ, θS, θF, π/64}]};

 axes = With[{mult = 1.2, color = LightGray},
   Axis[O, #[[1]], #[[2]], mult, color] & /@ {{𝟙x, "x"}, {𝟙y, "y"}, {𝟙z, "z"}}];
 range = 1.2 {{-0.1, 1}, {-0.1, 1}, {-0.1, 1}};
 {P, Q} =
  With[{R = 0.8, θ = 3 π/8, z = 0.8}, {CyltoCar[R, θ, z], CyltoCar[R, θ, 0]}];
 refLines = {Black, Line[{P, Q}], Line[{O, Q}], Point[P], Text[Stl["P"], 1.1 P],
   Text[Stl["z"], (P + Q)/2 + {0, 0.1, 0}], Text[Stl["R"], Q/2 + {0, 0, 0.1}]};
 ϕArc = With[{R = 0.4, θ = 3 π/8, z = 0}, {Black, Arrowheads[0.02],
    ArcArrow3D[R, 0, θ, 0], Text["θ", CyltoCar[0.7 R, θ/2, z]]}];
 G[1] = Graphics3D[{axes, refLines, ϕArc}, BoxRatios → Automatic,
   Boxed → False, Axes → None, PlotRange → range,
   ViewPoint → {2, 2, 1}, PlotLabel → Stl["cylindrical geometry"] ];
 G[2] = With[{R = 0.8}, ParametricPlot3D[{{R Cos[θ], R Sin[θ], z}}, {θ, 0, π/2},
    {z, 0, 1}, Mesh → False, ColorFunction → "TemperatureMap", PlotStyle → Opacity[.25],
    Axes → None, Boxed → False, ViewPoint → {2, 2, 1}, PlotRange → range]];
 Show[
  G[
   1],
  G[
   2]]]
```



Forces on a fluid element

```
Module[{O = {0, 0}, 𝟙x = {1, 0}, 𝟙y = {0, 1}, imageSize = 300,
  P, axes, OP, FC, FG, range, pointP, pointM, Axis2D, G},
 Axis2D[OO_, P_, lab_, mult_, color_] :=
  {Directive[color, Thick], Line[{OO, P}], Text[Stl[lab], mult P]};
 axes = With[{mult = 1.1, color = LightGray},
   Axis2D[O, #[[1]], #[[2]], mult, color] & /@ {{𝟙x, "R"}, {𝟙y, "z"}}];
 P = With[{r = 0.6, α = 1.5, ξ = π / 4}, {α r Cos[ξ], r Sin[ξ]}];
 pointP = {PointSize[0.02], Point[P]};
 pointM = {PointSize[0.03], Point[O]};
 FC = {Arrow[{P, P + {0.25, 0}}], Text[Stl["F_C"], (2 P + {0.25, 0})/2 + {0, 0.05}]};
 FG = {Arrow[{P, P/2}], Text[Stl["F_G"], (P + P/2)/2 + {0, -0.05}]};
 range = 1.2 {{-0.1, 1}, {-0.1, 1}};
 G[1] = With[{r = 0.6, α = 1.5},
   ParametricPlot[{{α r Cos[ξ], r Sin[ξ]}},
    {ξ, 0, π/2}, PlotStyle → Black, Axes → None, PlotRange → range,
    PlotLabel → Stl["forces on fluid element"], ImageSize → imageSize]];
 G[2] = Graphics[{axes, FG, FC, pointP, pointM, Text[Stl["M"], O + {0, -0.1}]},
   PlotRange → range, ImageSize → imageSize];
 Show[G[1], G[2]]]
```